MATH 2250-3
SPRINGS AND EARTHQUAKES
November 17, 2003

Your final Maple project for Math 2250 this semester is the Earthquake project on pages 437-438 of Edwards-Penney. The project is at our home page http://www.math.utah.edu/~korevaar/2250fall03/2250maple.html. In these notes we will work through the book examples from section 7.4, using illustrative Maple commands, as a warmup for your project work.

Let's start with example 1 on page 427 of Edwards-Penney, which we have also been working by hand. Initially it is an unforced system with two masses and two springs, as you can see from the description on page 427. We can write the system as **Mx''=Kx**, where **M** is the ''mass matrix'', **K** is the ''spring matrix'', and **x** is the displacement vector. Following the book's notation, we enter

```
> with(linalg):with(plots):with(DEtools): #tools
> M:=matrix([[2,0],[0,1]]);
  K:=matrix([[-150,50],[50,-50]]);
  A:=evalm(inverse(M)&*K);
```

$$M := \begin{bmatrix} 2 & 0 \\ 0 & 1 \end{bmatrix}$$

$$K := \begin{bmatrix} -150 & 50 \\ 50 & -50 \end{bmatrix}$$

$$A := \begin{bmatrix} -75 & 25 \\ 50 & -50 \end{bmatrix}$$

Then the system can also be written as **x''=Ax**, and the eigenvectors of A determine fundamental modes, and the corresponding negative eigenvalues are the (opposites) of the squares of the corresponding angular frequencies:

```
> eigenvects(A);
```

$$[-100, 1, \{[-1, 1]\}], [-25, 1, \{[1, 2]\}]$$

Therefore, the natural frequencies of this system are the 10 and 5, and the two fundamental modes correspond to the masses moving in opposite directions (with equal amplitudes and angular frequency 10) and in parallel directions (with amplitude ratio of two and angular frequency 5).

Now, let's consider the forced system with force vector equal to cos(wt)[0,50], i.e. the second mass is being forced periodically. In other words, the system **Mx''=Kx + F**, where F=cos(wt)[0,50] discussed on page 433. We follow the method described on that page to find a particular solution to the forced oscillation problem, of the form given by equation (31). The details of this computation are explained in example 3 of the text and in our handwritten notes and here is a Maple version:

```
> F0:=evalm(inverse(M)&*vector([0,50]));
      #The F0 in the normalized equation (30), page 433
  Iden:=array(1..2,1..2,identity);
      #the 2 by 2 identity matrix
  Aleft:=omega->evalm(A + omega^2*Iden);
      #the matrix function on the left side of (32)
  c:=omega->evalm(-inverse(Aleft(omega))&*F0);
       #the vector c(omega) in (32)
```

$$F0 := [0, 50]$$

$$Iden := \text{array}(identity, 1 .. 2, 1 .. 2, [\ ])$$

$$Aleft := \omega \to \text{evalm}(A + \omega^2\, Iden)$$

$$c := \omega \to \text{evalm}(-\text{`\&*`}(\text{inverse}(\text{Aleft}(\omega)), F0))$$

```
> c(omega); #see equation (35) page 433
```

$$\left[\frac{1250}{2500 - 125\,\omega^2 + \omega^4}, -\frac{50\,(-75 + \omega^2)}{2500 - 125\,\omega^2 + \omega^4}\right]$$

The vector c(w) above, times the oscillation cos(wt), is a particular solution to the forced oscillation problem we are considering. If we assume that our actual problem has a small amount of damping, then we expect that this particular solution is very close to the steady state solution to the damped problem. See the discussion on page 434. We can study resonance phenomena for these slightly damped problems by plotting the maximum amplitude of the steady state solutions to the undamped problems, much like you did in the Tacoma Narrows project. Use ‘‘norm, infinity’’ to measure this maximum amplitude:

```
> norm(c(omega),infinity);
```

$$\max\left(50\left|\frac{-75 + \omega^2}{2500 - 125\,\omega^2 + \omega^4}\right|, \frac{1250}{\left|2500 - 125\,\omega^2 + \omega^4\right|}\right)$$

```
> plot(norm(c(omega)),omega=0..15,y=0..15,
         numpoints=200,color=‘black‘);
```



This is the picture on page 434. It measures the maximum amount that one of the masses will be displaced from equilibrium, for the particular solution we derived. Notice the peaks are at angular frequency 5 and 10, reflecting the resonance which will occur at the natural frequencies.

We can get a plot of resonance as a function of period by recalling that 2*Pi/T=omega:

```
> res:=T->norm(c(2*Pi/T));
  plot(res(T),T=0.1..3,y=0..15,numpoints=200,color='black');
```

$$res := T \to \mathrm{norm}\left(\mathrm{c}\left(\frac{2\,\pi}{T}\right)\right)$$



COMMENTS FOR THE EARTHQUAKE PROJECT:

(1) Students are often confused by the forcing term in equation (2) of page 438, namely

```
> E*(omega)^2*cos(omega*t)*b;
```

$$E\,\omega^2\cos(\omega\,t)\,b$$

where b is the transpose of [1,1,1,1,1,1,1]. They ask, ‘‘how can the earthquake be forcing all seven stories, it seems like it's just shaking the bottom one.’’ Well, the students are correct, but so is Edwards-Penney. The authors talk about an ‘‘opposite inertial force’’ being the reason for this forcing term and here's one way to think about it: Think of the ground as the zeroth story. In the rest frame it is shaking with oscillation Ecos(wt). And so its acceleration is its second time derivative, namely -E*w^2*cos(wt). If you write down the inhomogeneous system of EIGHT second order DE's for the accelerations of stories zero thru seven, the forcing (well, accelerating) term is -E*w^2*cos(wt)*[1,0,0,0,0,0,0,0], as you would expect. Call the solution 8-vector to this system **y**(t), then see what the shaking looks like to someone on the ground by letting **x**(t)=**y**(t)-E*cos(wt)*[1,1,1,1,1,1,1,1]. Then the zeroth story component of x(t) will be identically zero, and the other seven components will satisfy equation (2) on the bottom of page 303, exactly as the authors claim.

(2) For large matrices the eigenvect command won't work well unless Maple knows you want a decimal approximation: If all entries are rational numbers (expressed without decimal points), Maple tries to find the eigenvalues and eigenvectors algebraically and exactly, instead of numerically, and often fails. The way around this is to either ask for evalf(eigenvects(A)), or to Make sure at least one of your matrix entries has a decimal point in it.